Multi-headed Lattice Green Function (N = 4, M = 2)
Find Minimal REC

In[∘]:=
```
NN = 4;
MM = 2;
```

Recall some basic definitions in the paper:

$$P_{M,N}(z) := \frac{1}{(2\pi)^N} \int_{-\pi}^{\pi} \dots \int_{-\pi}^{\pi} \frac{1}{1 - \frac{z}{\binom{N}{M}} \sigma_M(\cos\theta_1, \dots, \cos\theta_N)} d\theta_1 \dots d\theta_N$$

$$R_{M,N}(z) := P_{M,N}\left(2^M \binom{N}{M} z\right) \text{ and } R_{M,N}(z) = \sum_{n\geq 0} r_{M,N}(n) z^n$$

Also, for M odd or $M = N$, we always have $r(2n+1) = 0$. Hence, define
$\tilde{r}_{M,N}(n) := r_{M,N}(2n)$ and $\tilde{R}_{M,N}(z) := \sum_{n\geq 0} \tilde{r}_{M,N}(n) z^n = \sum_{n\geq 0} r_{M,N}(2n) z^n$

Our goal is to find:

Case 1. *M* even and *M* ≠ *N*:
- minimal recurrences (REC) for $r(n)$.

Case 2. *M* odd or *M* = *N*:
- minimal recurrences (REC) for $\tilde{r}(n)$.

Command: UnrollRecurrence

Generate a sequence from recurrence & initial values (Koutschan's implementation).

In[∘]:=
```
(* given a recurrence rec in f[n], compute the values {f[0],f[1],...,f[bound]}
 where inits are the initial values
 {f[0],...,f[d-1]} with d being the order of the recurrence *)
Clear[UnrollRecurrence];
UnrollRecurrence[rec1_, f_[n_], inits_, bound_] :=
  Module[{i, x, vals = inits, rec = rec1},
   If[Head[rec] =!= Equal, rec = (rec == 0)];
   rec = rec /. n → n - Max[Cases[rec, f[n + a_.] ⧴ a, Infinity]];
   Do[
    AppendTo[vals, Solve[rec /. n → i /. f[i] → x /. f[a_] ⧴ vals[[a + 1]], x][[1, 1, 2]]];
    , {i, Length[inits], bound}];
   Return[vals];
  ];
```

Load RISC packages.

In[∘]:=
```
<< RISC`HolonomicFunctions`
<< RISC`Asymptotics`
<< RISC`Guess`
```

HolonomicFunctions Package version 1.7.3 (21-Mar-2017)
written by Christoph Koutschan
Copyright Research Institute for Symbolic Computation (RISC),
Johannes Kepler University, Linz, Austria

```
--> Type  ?HolonomicFunctions  for help.
```

Asymptotics Package version 0.3
written by Manuel Kauers
Copyright Research Institute for Symbolic Computation (RISC),
Johannes Kepler University, Linz, Austria

Package GeneratingFunctions version 0.9 written by Christian Mallinger
Copyright Research Institute for Symbolic Computation (RISC),
Johannes Kepler University, Linz, Austria

Guess Package version 0.52
written by Manuel Kauers
Copyright Research Institute for Symbolic Computation (RISC),
Johannes Kepler University, Linz, Austria

We start by importing known ODE for $R(z)$.

Note that the ODE in Koutschan (2013, p. 9, Thm 1) is for $P(z) = R\left(z \Big/ \binom{N}{M} 2^{M}\right)$.

In[∘]:=
```
ODEDiv2 =
  ToOrePolynomial[12 * z * (256 + 632 * z + 702 * z^2 + 382 * z^3 + 98 * z^4 + 9 * z^5) * P[z] + 12 *
      (-384 + 224 * z + 3716 * z^2 + 7633 * z^3 + 6734 * z^4 + 2939 * z^5 + 604 * z^6 + 45 * z^7) *
      Derivative[1][P][z] + 6 * z * (-5376 - 5248 * z + 11 080 * z^2 + 25 286 * z^3 + 19 898 * z^4 +
        7432 * z^5 + 1286 * z^6 + 81 * z^7) * Derivative[2][P][z] + 2 * z^2 * (4 + 3 * z) *
      (-3456 - 2304 * z + 3676 * z^2 + 4920 * z^3 + 2079 * z^4 + 356 * z^5 + 21 * z^6) *
      Derivative[3][P][z] + (-1 + z) * z^3 * (2 + z) * (3 + z) * (6 + z) * (8 + z) * (4 + 3 * z)^2 *
      Derivative[4][P][z] /. {Derivative[k_][P][z] → Der[z]^k} /. {P[z] → 1}];
```

Process the data.

Write the ODE in terms of the operators D and θ.

In[∘]:=
```
ODENormalizedinD = NormalizeCoefficients[DFiniteSubstitute[{ODEDiv2},
     {z → w * 2^MM * Binomial[NN, MM]}, Algebra → OreAlgebra[Der[w]]][[1]]];
```

In[∘]:=
```
ODENormalizedinTheta =
  NormalizeCoefficients[ChangeOreAlgebra[w ** ODENormalizedinD, OreAlgebra[Euler[w]]]];
```

Then transform the above to a REC for $r(n)$ and write it explicitly.

```
In[◦]:= RECNormalizedinS = NormalizeCoefficients[DFiniteDE2RE[{ODENormalizedinD}, {w}, {α}][[1]]];

In[◦]:= RECNormalizedinSOrder = OrePolynomialDegree[RECNormalizedinS, S[α]]

Out[◦]= 7

In[◦]:= ClearAll[Seq];
        SeqNormalized = ApplyOreOperator[RECNormalizedinS, Seq[α]];
```

Compute the initial values of $r(n)$.

```
In[◦]:= MAX = RECNormalizedinSOrder;
        ClearAll[a];

        SeriesIni = ApplyOreOperator[ODENormalizedinTheta, Sum[a[n] w^n, {n, 0, MAX}]];

        SeriesIniSol = Solve[Join[Table[Coefficient[SeriesIni, w, i] == 0, {i, 1, MAX}], {a[0] == 1}],
          Table[a[i], {i, 0, MAX}]]

        SeqListIni = Table[SeriesIniSol[[1, k, 2]], {k, 1, Length@SeriesIniSol[[1]]}]

        seq[n_] := SeqListIni[[n + 1]];

Out[◦]= {{a[0] → 1, a[1] → 0, a[2] → 24, a[3] → 192,
          a[4] → 3384, a[5] → 51840, a[6] → 911040, a[7] → 16369920}}

Out[◦]= {1, 0, 24, 192, 3384, 51840, 911040, 16369920}
```

Generate a list of $r(n)$.

```
In[◦]:= Bound = 200;

        SeqList = UnrollRecurrence[SeqNormalized, Seq[α], SeqListIni, Bound];

        seq[n_] := SeqList[[n + 1]];
```

Guess a Minimal REC for $r(n)$.

SeqfromRECGuess gives the REC in Theorem 6.1! (To be displayed at the end of this notebook)

REC: Order 5

ODE: Order 11, Degree 5

```
In[◦]:= ClearAll[Seq];
        RECGuess = GuessMinRE[Take[SeqList, 200], Seq[α]];
        RECGuessinS = NormalizeCoefficients[ToOrePolynomial[RECGuess /. {Seq[k_] → S[α]^(k-α)}]];

In[◦]:= RECGuessinSOrder = OrePolynomialDegree[RECGuessinS, S[α]]

Out[◦]= 5

In[◦]:= ODEfromRECGuessinD = NormalizeCoefficients[DFiniteRE2DE[{RECGuessinS}, {α}, {z}][[1]]];
```

In[∘]:=
```
ODEfromRECGuessinTheta =
  NormalizeCoefficients[ChangeOreAlgebra[z ** ODEfromRECGuessinD, OreAlgebra[Euler[z]]]];
```

In[∘]:=
```
ODEfromRECGuessinThetaOrder = OrePolynomialDegree[ODEfromRECGuessinTheta, Euler[z]]
```

Out[∘]= 11

In[∘]:=
```
ODEfromRECGuessinThetaDegree =
 Max[Exponent[OrePolynomialListCoefficients[ODEfromRECGuessinTheta], z]]
```

Out[∘]= 5

We may also write this REC explicitly.

In[∘]:=
```
ClearAll[Seq];
SeqfromRECGuess = ApplyOreOperator[RECGuessinS, Seq[α]];
```

In[∘]:=
```
SeqfromRECGuessList =
  UnrollRecurrence[SeqfromRECGuess, Seq[α], Take[SeqList, RECGuessinSOrder], 200];
```

Prove the minimal REC for *r*(*n*).

In[∘]:=
```
RECCompare = DFinitePlus[{RECNormalizedinS}, {RECGuessinS}][[1]];
```

In[∘]:=
```
RECCompareOrder = OrePolynomialDegree[RECCompare, S[α]]
```

Out[∘]= 7

In[∘]:=
```
CheckNum = RECCompareOrder + 20;
Take[SeqList, CheckNum] - Take[SeqfromRECGuessList, CheckNum]
```

Out[∘]= {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}

Display the REC in Theorem 6.1

In[∘]:=
```
Collect[Expand[-SeqfromRECGuess], Seq[_]]
```

Out[∘]=
$$\left(287649792 + 787304448\,\alpha + 833891328\,\alpha^2 + 441427968\,\alpha^3 + 123641856\,\alpha^4 + 17418240\,\alpha^5 + 967680\,\alpha^6\right) \mathrm{Seq}[\alpha] + \left(708258816 + 1417457664\,\alpha + 1162038528\,\alpha^2 + 498714624\,\alpha^3 + 117891072\,\alpha^4 + 14515200\,\alpha^5 + 725760\,\alpha^6\right) \mathrm{Seq}[1+\alpha] + \left(379157760 + 643100256\,\alpha + 452539152\,\alpha^2 + 168897600\,\alpha^3 + 35209440\,\alpha^4 + 3880800\,\alpha^5 + 176400\,\alpha^6\right) \mathrm{Seq}[2+\alpha] + \left(55519056 + 84088296\,\alpha + 52997120\,\alpha^2 + 17786040\,\alpha^3 + 3351200\,\alpha^4 + 336000\,\alpha^5 + 14000\,\alpha^6\right) \mathrm{Seq}[3+\alpha] + \left(-638976 - 904864\,\alpha - 533288\,\alpha^2 - 167156\,\alpha^3 - 29341\,\alpha^4 - 2730\,\alpha^5 - 105\,\alpha^6\right) \mathrm{Seq}[4+\alpha] + \left(-345000 - 451000\,\alpha - 244675\,\alpha^2 - 70540\,\alpha^3 - 11402\,\alpha^4 - 980\,\alpha^5 - 35\,\alpha^6\right) \mathrm{Seq}[5+\alpha]$$